High Accuracy Behavioral Modeling for Frequency and Time Domain Simulations

IBIS Summit Meeting at DAC 2001

Hilton Hotel, Las Vegas, Nevada

June 21, 2001

Arpad Muranyi

arpad.muranyi@intel.com

Signal Integrity Engineering

Intel Corporation

Outline

Part I – Problem statement

- Why frequency domain analysis?
- Resonant circuit example

Part II – Frequency response of an IBIS model

- The impedance of an IBIS model
- Facts and problem statement

Part III – New modeling proposal

- Some theory and complex algebra
- Transfer functions and Laplace elements
- HSPICE examples
- Introducing the FD/TD behavioral model
- More HSPICE examples
- Conclusion

Part I

Introduction

Why frequency domain analysis?

Some times it is easy to tell that there is a signal quality problem by looking at the waveforms (in time domain), but often it is very easy to miss problems or it may take too long to find them.

We need a methodology that can reliably and quickly tell us whether there is a problem and what it is.

- **The waveforms of time domain simulations may change when we run the simulation at different frequencies even if we don't change anything in the circuit (due to reflections and resonance)**
  - **this frequency dependency can be studied much easier with frequency domain analysis to uncover problems**
  - **FD simulations run much faster than TD simulations, so more work can be done in less time**

Once we are done optimizing for signal quality in the frequency domain we still need to run simulations in the time domain to find best/worst timings for the design.

The bottom line is time

The timing of our designs is determined by the shape of the waveforms.

The shape of the waveforms, however, are very strongly effected by the resonance effects.

Eliminating resonance results in more precise, stable and predictable timings. So our goal is to find and remove resonances.

In order to be able to do that we need buffer models which are accurate at all frequencies.

What makes these waveforms so unpredictable?

? ? ? ? ?



Waveforms without resonance are predictable



The circuit of the previous two sets of waveforms



L = 10 nH
C = 10 pF

underdamped R = 5 Ω **or** R = 63.25 Ω **critically damped**

f_{clock} = 100, 375, 500, 750 MHz

Resonant ω_0 = 503.3 MHz **NOT resonant**

$$s_{1,2} = -\frac{R}{2L} \pm \sqrt{\left(\frac{R}{2L}\right)^2 - \frac{1}{LC}}$$

$$\alpha = \frac{R}{2L} = \omega_0 = \sqrt{\frac{1}{LC}}$$

$$s_{1,2} = -\alpha \pm \sqrt{\alpha^2 - \omega_0^2}$$

$$R = 2\sqrt{\frac{L}{C}} = 2\sqrt{\frac{10\ nH}{10\ pF}} = 2\sqrt{1000} = 63.25\ \Omega$$

Part II

Frequency response of an IBIS model

Looking into an IBIS buffer - Z_{11}



$Z_{11} = V_1/I_1$

Where Z_{11} is a complex quantity (Real + Imaginary)

Frequency response at 0.1 V DC bias



Frequency response at 2.0 V DC bias



Desktop Platforms GROUP

Facts and problem statement

- Voltage dependent impedance (Z_{11} vs. bias)
  - this is the steady state DC behavior
  - IBIS uses IV curves to describe this
  - good match (at very low frequencies)
- Frequency dependent impedance
  - we use C_comp to describe (part of) it
  - C_comp matches a single parallel RC circuit well
  - series-parallel RC combinations not accurate
  - cannot model voltage (bias) dependency
- Time dependent impedance
  - this is the transient (switching) behavior
  - IBIS uses Vt curves to describe this
  - need to find a method to analyze AC response during transients

Part III

New modeling proposal

Some theory



R C

Complex number = Real part + Imaginary part

$$Z = R + jX = R \parallel 1/j\omega C$$

$$Y = G + jB = \underbrace{G + j\omega C}_{\text{(for a parallel RC circuit)}}$$



R(V) C(V)

To express this as a function of voltage, we can write

$$Z(V) = R(V) \parallel 1/j\omega C(V)$$

$$Y(V) = G(V) + j\omega C(V)$$

Transfer functions and the Laplace element

The Y_{11} transfer function of a resistor:

$$H(s) = 1/R$$

The Y_{11} transfer function of a capacitor:

$$H(s) = sC$$

The Y_{11} transfer function of a parallel RC circuit:

$$H(s) = \frac{1}{R} + sC = \frac{1 + sCR}{R}$$

In HSPICE this can be modeled with a Laplace element the following way:




```
G1  node1 node2  Laplace  node1 node2  1 'R*C' / 'R'
```

An undocumented(?) HSPICE feature

The same transfer function can also be expressed with parameters who's value depends on another variable. The Y_{11} transfer function of the parallel RC circuit as a function of voltage is:

$$H(s,V) = \frac{1}{R(V)} + sC(V) = \frac{1 + sC(V)R(V)}{R(V)}$$

(This works in HSPICE up to a few dependent terms).




```
G1  n1 n2  Laplace  n1 n2
+  1 'R*V(n1,n2)*C*V(n1,n2)' / 'R*V(n1,n2)'
```

Proving that it works

```
Laplace element coefficients as a function of voltage
*****************************************************
.TRAN 1.0e-12 15.0e-9
.OPTIONS ACCT POST=1
*************************************
.param Rval = 200
.param Cval = 10.0pF
*************************************
.IC V(RC1) = 1
R1     RC1   0   R=Rval
C1     RC1   0   C=Cval
*
.IC V(RC2) = 1
Rd2    RC2   0   R=1e+6
G2     RC2   0   LAPLACE   RC2   0
+ 1 'Rval*Cval' / Rval
*
.IC V(RC3) = 1
R3     RC3   0   R='Rval*(1/V(RC3))'
C3     RC3   0   C=Cval
*
.IC V(RC4) = 1
Rd4    RC4   0   R=1e+6
G4     RC4   0   LAPLACE   RC4   0
+ 1 'Rval*(1/V(RC4))*Cval' / 'Rval*(1/V(RC4))'
*****************************************************
.END
*****************************************************
```



Revisiting complex algebra

$$\underbrace{(a + j0)}_{Z_{DC}} + \underbrace{(b + jc)}_{Z_{AC}} = \underbrace{(a+b) + jc}_{Z_{measured}}$$

IV curve | **AC parasitics** | **Total response**



Total response DC + AC

PWL current source | **Laplace current source**

$$Z_{measured} - Z_{DC} = Z_{AC}$$

Since Z_{DC} (IV curve) is a real number, it does not have an imaginary part, and $Im(Z_{AC}) = Im(Z_{measured})$.

Also, because Z_{DC} is a frequency independent number (i.e. constant), $Re(Z_{AC}) = Re(Z_{measured}) - k$ (which is a simple shift)

Medium complexity example with Mathcad

Two parallel RCs in series

$R1 := 10 \qquad C1 := 1.0 \cdot 10^{-12}$

$R2 := 90 \qquad C2 := 9.0 \cdot 10^{-12}$

$f := 0, 1 \cdot 10^{6} .. 2 \cdot 10^{9}$

$\omega(f) := 2 \cdot \pi \cdot f$

$s(f) := 0 + i \cdot \omega(f)$

$$H(f) := \frac{1}{\dfrac{1}{\dfrac{1}{R1} + s(f) \cdot C1} + \dfrac{1}{\dfrac{1}{R2} + s(f) \cdot C2}}$$

or

$$H(f) := \frac{1 + s(f) \cdot (C1 \cdot R1 + C2 \cdot R2) + s(f)^2 \cdot C1 \cdot R1 \cdot C2 \cdot R2}{R1 + R2 + s(f) \cdot (R1 \cdot C2 \cdot R2 + C1 \cdot R1 \cdot R2)}$$



if $Hdc(f) := \dfrac{1}{R1 + R2}$ **then** $Hac(f) := H(f) - Hdc(f)$

$$Hac(f) := \frac{\left(C1 \cdot R1^2 \cdot C2 \cdot R2 + C1 \cdot R1 \cdot C2 \cdot R2^2\right) \cdot s(f)^2 + \left(C1 \cdot R1^2 + C2 \cdot R2^2\right) \cdot s(f)}{\left(R1^2 \cdot C2 \cdot R2 + R1 \cdot C2 \cdot R2^2 + C1 \cdot R1^2 \cdot R2 + C1 \cdot R1 \cdot R2^2\right) \cdot s(f) + R1^2 + 2 \cdot R1 \cdot R2 + R2^2}$$

$n0 := 0 \qquad n0 = 0$

$n1 := C1 \cdot R1^2 + C2 \cdot R2^2 \qquad n1 = 7.3 \cdot 10^{-8}$

$n2 := C1 \cdot R1^2 \cdot C2 \cdot R2 + C1 \cdot R1 \cdot C2 \cdot R2^2 \qquad n2 = 8.1 \cdot 10^{-19}$

$d0 := R1^2 + 2 \cdot R1 \cdot R2 + R2^2 \qquad d0 = 1 \cdot 10^{4}$

$d1 := R1^2 \cdot C2 \cdot R2 + R1 \cdot C2 \cdot R2^2 + C1 \cdot R1^2 \cdot R2 + C1 \cdot R1 \cdot R2^2 \qquad d1 = 9 \cdot 10^{-7}$

$d2 := 0 \qquad d2 = 0$

Real and Imaginary part plots from Mathcad





HSPICE netlist segment for Mathcad example

```
**********************************************************************
.param Bias= 0.0
.param C1 = 1.0pF
.param C2 = 9.0pF
.param R1 = 10
.param R2 = 90
*
.param n0 = 0
.param n1 = 'C1*pow(R1,2)+C2*pow(R2,2)'
.param n2 = 'C1*pow(R1,2)*C2*R2+C1*R1*C2*pow(R2,2)'
.param d0 = 'pow(R1,2)+2*R1*R2+pow(R2,2)'
.param d1 = 'pow(R1,2)*C2*R2+R1*C2*pow(R2,2)+C1*pow(R1,2)*R2+C1*R1*pow(R2,2)'
.param d2 = 1e-21                        $ 1e-21 to keep HSPICE happy
**********************************************************************
Vac1  RC1    0     AC= 1  DC= Bias
*
R1a   RC1    RC1a  R= R1
C1a   RC1    RC1a  C= C1
R1b   RC1a   0     R= R2
C1b   RC1a   0     C= C2
**********************************************************************
Vac2  RC2    0     AC= 1  DC= Bias
G2a   RC2    0     CUR='V(RC2)/100'    $ Parallel R
*---------------------------------------------------------------------
G2b   RC2    0     LAPLACE  RC2   0
*                                      $ Two || RCs in series minus DC
+ n0 n1 n2 / d0 d1 d2
**********************************************************************
.END
**********************************************************************
```

Frequency domain response with HSPICE



Time domain response with HSPICE (.IC = 5 V)



HSPICE netlist segment of real buffer model

```
*********************************************************************
Vac2       RC2            0      AC=Vac     DC= Bias
X2a        RC2            0      FET_IV
*-------------------------------------------------------------------
G2b        0              RC2    LAPLACE    RC2   0
+ '-4.6156E-02 \\
   +2.1192E-02*V(RC2) \\
   +9.2658E-04*pow(V(RC2),2) \\
   -1.1509E-03*pow(V(RC2),3) \\
   +2.1795E-06*pow(V(RC2),4) \\
   +1.8981E-05*pow(V(RC2),5)'
+ '-8.6576E-12 \\
   -2.4567E-11*V(RC2) \\
   +2.7304E-11*pow(V(RC2),2) \\
   -1.0631E-11*pow(V(RC2),3) \\
   +1.8243E-12*pow(V(RC2),4) \\
   -1.1699E-13*pow(V(RC2),5)'
+ '-1.9814E-22 \\
   -8.1507E-22*V(RC2) \\
   +6.4066E-22*pow(V(RC2),2) \\
   -2.6783E-22*pow(V(RC2),3) \\
   +5.5096E-23*pow(V(RC2),4) \\
   -4.3215E-24*pow(V(RC2),5)' /
+ 1
+ '+1.5954E-10 \\
   +6.5574E-10*V(RC2) \\
   -4.8825E-10*pow(V(RC2),2) \\
   +1.8850E-10*pow(V(RC2),3) \\
   -3.6452E-11*pow(V(RC2),4) \\
   +2.6923E-12*pow(V(RC2),5)'
+ '+7.8650E-23 \\
   +3.2570E-22*V(RC2) \\
   -2.5402E-22*pow(V(RC2),2) \\
   +1.0774E-22*pow(V(RC2),3) \\
   -2.2593E-23*pow(V(RC2),4) \\
   +1.8119E-24*pow(V(RC2),5)'
*********************************************************************
```

Frequency response of a real buffer with HSPICE



dV/dI and dR/dV



R1 = V1 / I1

R2 = V2 / I2

but

$$\frac{V2-V1}{I2-I1} \neq R2-R1$$

or

$$\frac{dV}{dI} \neq \frac{dR}{dV}$$

In .AC analysis SPICE uses the slope of the IV curves for transistors, diodes and PWL sources. However, an IV curve implemented as a voltage controlled resistor (VCR) will produce delta R in a .AC sweep.

For this reason we need to use PWL sources to implement IV curves (in SPICE) for behavioral models.

AC sweep of FET, I-V and R-V models

The real part of the RV model at DC is incorrect!

The real part of the IV model at DC is correct.



The AC component at higher bias voltages is incorrect as explained before.

Issues / work in progress

- **TRAN mode simulations of real buffer is still in progress**
- **Numerical issues make it difficult to achieve a working model**
  - **the accuracy of these experiments was limited by the order of curve fitting function (due to the available space in the HSPICE Laplace element)**
  - **not may tools can curve fit to a complex number data table**
    - **HSPICE optimization doesn't converge well**
  - **second curve fitting for voltage dependency**
    - **could be more accurate if done in a single process as a multi dimensional problem**
  - **H(s) and table format for Laplace element did not converge in .TRAN yet for a real buffer behavioral model**
    - **could be due to numerical stability issues (denominator must be higher order than numerator), or the range of exponents is too wide for HSPICE**
    - **rearranging transfer function to have higher order denominator may get this to work**
  - **could try pole-zero format**
  - **these problems may not be an issue if a tool was designed to do this**
- **Transient behavior of AC components need to be studied more**
  - **extending the concept to multi dimensional transfer functions (time variant, voltage dependent) could account for transients**
  - **could possibly use the V-t curves or similar information to control transient behavior of the AC components of the model, but more research is needed to understand this**

Conclusions

- **We need to consider only one additional element for the IBIS-X specification to achieve a highly accurate FD/TD behavioral model**
  - **this element would replace the existing C_comp parameter**
  - **this element needs to interpret some sort of a transfer function**
  - **various formats could be supported, table driven, transfer function (equation) driven, or pole-zero driven**
  - **various types of data could be supported, Z, Y, S-parameters, and/or voltage and current (each of which is complex)**
- **Complex number capability is needed to make this work**
- **The concept shown here using a one-port circuit can be extended to a special multi terminal port representing all nodes of the buffer**
  - **need at least two terminals for the power and ground rails**
  - **need one or two terminals for single ended or differential buffers**
  - **the complex impedance as a function of output and supply voltage, time, temperature, process, etc. between each of these terminals can be described according to the concept shown here**